Q:

Find word in part of cell text, return column number: Excel VBA

My goal is to implement a search function in a user form in Excel.
It should be able to take a word entered in a TextBox and go through every column in a specific row and search for the word. Then it should return the column where it was first discovered. 
With my code below, it will look only at exact matches of the cell content. But if the cell value is for example: "Lorem ipsum, dolor sitamet" - I want to be able to search for "dolor". 
Here is my code:
Dim rFind As Range     
With Range("D1:D100")
    Set rFind = .Find(What:=TextBox13.Value, LookAt:=xlWhole, MatchCase:=False, SearchFormat:=False)
    If Not rFind Is Nothing Then
        MsgBox rFind.Column
        MsgBox rFind.Row
    End If
End With

A:

Just change LookAt:=xlWhole to LookAt:=xlPart
Docs: Range.Find Method (Excel)

You can also use an ampersand (&) to tidy up the returned result, it concatenates strings:
"Col: " & rFind.Column & ", Row: " & rFind.Row
' For example, gives the string "Col: 1, Row: 1" if found in A1

You say you want to search in "every column in a specific row" but you're looking in "the first 100 rows in a specific column"?! Not sure whether that's a typo or if your code is wrong. Either way, you could use the Rows or Columns objects. These should be used with fully qualifying your range, i.e. saying where it is.
ThisWorkbook.Sheets("Sheet1").Rows(1)      ' Range of entire first row
ThisWorkbook.Sheets("Sheet1").Columns(1)   ' Range of entire first column
ThisWorkbook.Sheets("Sheet1").Columns("A") ' Equivalent, range of entire first column

Rewrite:
Dim rFind As Range     
With ThisWorkbook.Sheets("Sheet1").Columns("D")
    Set rFind = .Find(What:=TextBox13.Value, LookAt:=xlPart, MatchCase:=False, SearchFormat:=False)
    If Not rFind Is Nothing Then
        MsgBox "Col: " & rFind.Column & ", Row: " & rFind.Row
    End If
End With

